
2 Correction required for 6.3.3 (IMS enabled on UTRA)
2.1 f_TC_6_3_3_UTRAN 
	Object name 
	f_TC_6_3_3_UTRAN

	Reason for change
	The function f_UTRAN_GMMOnly_Attach(utran_Cell5); is called for the steps 2 to 13 of the expected test sequence. This function call does not take into account the condition when the UE is IMS enabled. This needs to be handled.

After Steps 2 to 13 on UTRAN are done, the control is transferred to the EUTRA PTC when the user manually selects the EUTRA CSG cell.  However, in the current implementation, there is no mechanism to notify the EUTRA PTC that a PDP context has already been established on UTRAN, which can cause issues for the expected signalling messages on EUTRAN. This mechanism needs to be implemented. Please see screenshot

	Summary of change
	Instead of calling f_UTRAN_GMMOnly_Attach(utran_Cell5);, called f_UTRAN_GMMOnly_Attach(utran_Cell5,true);.

Called the function f_IRAT_SendCoOrd to inform the EUTRA PTC about the PDP context status on UTRAN (Whether there is a PDP context activated on UTRAN or not). 

Please see screenshot.  

	Module
	CSG_TestcasesUG_UTRAN.ttcn

	Label
	WA#


Before:
	function f_TC_6_3_3_UTRAN() runs on UTRAN_IRAT_PTC
  {
    var IRAT_CoOrd_SysInfo_Type v_EUTRASysInfo;
    var UtraServingCellPriorityValue_Type v_UtraPriorityValue := 5;
    var EUTRA_MeasurementBandwidth v_EUTRA_MeasurementBandwidth := f_UTRAN_ConvertIntegerToEUTRA_MeasurementBandwidth(f_ConvertDL_BandwidthPIXITToInteger());
    timer t_WaitTime := tsc_IdleMode_GenericTimer_UTRAN;
    //Initialise all cells
    f_UTRAN_Init(EUTRA_UTRAN);
    //Set RAC of cell 7 to different from RAC of cell 5 according to Pre-test condition: "Cell 5 and Cell 7 have different Routing Area."
    f_UTRAN_CellInfo_SetRAC(utran_Cell7, tsc_RAC_2);
    //Send the cell info to the EUTRA component
    f_UTRAN_Send_IR_SysInfoToEUTRA (utran_Cell5);
    //Get Cell Parameters for EUTRA cell
    v_EUTRASysInfo := f_IRAT_WaitForCoOrd_SysInfo (EUTRA);
    //Update SIB19 of cell 5 according to Table 6.2.3.3.3.3-1
    f_UTRAN_SysInfo_SetSIB19(utran_Cell5,
                             cs_SIB19_UtraAndOneEutraEntry(cs_Utra_PriorityInfoList_Def(v_UtraPriorityValue),
                                                           cs_Eutra_FreqAndPriList_OneEntry_Def(v_EUTRASysInfo.Eutra[0].Arfcn,
                                                                                                v_EUTRA_MeasurementBandwidth)));
    f_UTRAN_SysInfo_SetSIB19(utran_Cell7,
                             cs_SIB19_UtraAndOneEutraEntry(cs_Utra_PriorityInfoList_Def(v_UtraPriorityValue),
                                                           cs_Eutra_FreqAndPriList_OneEntry_Def(v_EUTRASysInfo.Eutra[0].Arfcn,
                                                                                                v_EUTRA_MeasurementBandwidth)));
    //Create and configure Cell 5 & 7
    //Cell started switched on with default power level, -60 and max attenuation
    f_UTRAN_SS_CreateCellFACH (utran_Cell5);
    f_UTRAN_SS_CreateCellFACH (utran_Cell7);
    //Start sending System Information
    f_UTRAN_SendDefSysInfo (utran_Cell5);
    f_UTRAN_SendDefSysInfo (utran_Cell7);
    // Wait for CSG list to be emptied in LTE side
    f_IRAT_WaitForCoOrd_Trigger(EUTRA); //@sic R5s130922 sic@
    //Set power levels according to T0
    f_UTRAN_SetCellPower(utran_Cell5, -60, -62);
    f_UTRAN_TestBody_Set (true);
    //@siclog "Step 1" siclog@
    //The UE is switched on.
    f_UT_SwitchOnUE (UT, false);
    //@siclog "Step 2-13" siclog@
    //Perform registration procedure    
    f_UTRAN_GMMOnly_Attach (utran_Cell5);
    //Since Authentication parameters have been updated in the registration process, send them to EUTRA PTC
    f_UTRAN_SendAuthCoOrdParameters(EUTRA);
    f_IRAT_SendCoOrd(EUTRA, cms_IRAT_OctetData(f_UTRAN_MobileInfo_EUTRACapability_Get()));
    f_UTRAN_InterRAT_InitialiseAuthParams();
<<SKIPPED CODE>>


After:
	function f_TC_6_3_3_UTRAN() runs on UTRAN_IRAT_PTC
  {
    var IRAT_CoOrd_SysInfo_Type v_EUTRASysInfo;
    var UtraServingCellPriorityValue_Type v_UtraPriorityValue := 5;
    var EUTRA_MeasurementBandwidth v_EUTRA_MeasurementBandwidth := f_UTRAN_ConvertIntegerToEUTRA_MeasurementBandwidth(f_ConvertDL_BandwidthPIXITToInteger());
    timer t_WaitTime := tsc_IdleMode_GenericTimer_UTRAN;
    //Initialise all cells
    f_UTRAN_Init(EUTRA_UTRAN);
    //Set RAC of cell 7 to different from RAC of cell 5 according to Pre-test condition: "Cell 5 and Cell 7 have different Routing Area."
    f_UTRAN_CellInfo_SetRAC(utran_Cell7, tsc_RAC_2);
    //Send the cell info to the EUTRA component
    f_UTRAN_Send_IR_SysInfoToEUTRA (utran_Cell5);
    //Get Cell Parameters for EUTRA cell
    v_EUTRASysInfo := f_IRAT_WaitForCoOrd_SysInfo (EUTRA);
    //Update SIB19 of cell 5 according to Table 6.2.3.3.3.3-1
    f_UTRAN_SysInfo_SetSIB19(utran_Cell5,
                             cs_SIB19_UtraAndOneEutraEntry(cs_Utra_PriorityInfoList_Def(v_UtraPriorityValue),
                                                           cs_Eutra_FreqAndPriList_OneEntry_Def(v_EUTRASysInfo.Eutra[0].Arfcn,
                                                                                                v_EUTRA_MeasurementBandwidth)));
    f_UTRAN_SysInfo_SetSIB19(utran_Cell7,
                             cs_SIB19_UtraAndOneEutraEntry(cs_Utra_PriorityInfoList_Def(v_UtraPriorityValue),
                                                           cs_Eutra_FreqAndPriList_OneEntry_Def(v_EUTRASysInfo.Eutra[0].Arfcn,
                                                                                                v_EUTRA_MeasurementBandwidth)));
    //Create and configure Cell 5 & 7
    //Cell started switched on with default power level, -60 and max attenuation
    f_UTRAN_SS_CreateCellFACH (utran_Cell5);
    f_UTRAN_SS_CreateCellFACH (utran_Cell7);
    //Start sending System Information
    f_UTRAN_SendDefSysInfo (utran_Cell5);
    f_UTRAN_SendDefSysInfo (utran_Cell7);
    // Wait for CSG list to be emptied in LTE side
    f_IRAT_WaitForCoOrd_Trigger(EUTRA); //@sic R5s130922 sic@
    //Set power levels according to T0
    f_UTRAN_SetCellPower(utran_Cell5, -60, -62);
    f_UTRAN_TestBody_Set (true);
    //@siclog "Step 1" siclog@
    //The UE is switched on.
    f_UT_SwitchOnUE (UT, false);
    //@siclog "Step 2-13" siclog@
    //Perform registration procedure    
    f_UTRAN_GMMOnly_Attach (utran_Cell5, true/*WA#6_3_3*/);
    //Since Authentication parameters have been updated in the registration process, send them to EUTRA PTC
    f_UTRAN_SendAuthCoOrdParameters(EUTRA);
    f_IRAT_SendCoOrd(EUTRA, cms_PDP_Established(vc_UTRAN_Global.PDPEstEarlier));//WA#6_3_3
    f_IRAT_SendCoOrd(EUTRA, cms_IRAT_OctetData(f_UTRAN_MobileInfo_EUTRACapability_Get()));
    f_UTRAN_InterRAT_InitialiseAuthParams();
<<SKIPPED CODE>>


2.2 f_TC_6_3_3_EUTRA 
	Object name 
	f_TC_6_3_3_EUTRA

	Reason for change
	After preamble on UTRAN is done, the control is transferred to the EUTRA PTC which notifies whether a PDP context is established on UTRAN. However, in the current implementation, there is no mechanism to receive this notification. This needs to be implemented.
In case the testcase is being run with an IMS enabled device, this function needs to be provided to the function f_EUTRA_IdleMode_TAU_UponMobility_FromOtherRAT so that the correct path in TTCN can be followed for IMS enabled device. 

Also, based upon whether the testcase is run with an IMS enabled device, the call to function f_EUTRA_TrackingAreaUpdateFromAnotherRAT needs to be passed different set of parameters. 

This also needs to be done. 

	Summary of change
	Created a variable v_PDPEstEarlier and assigned to this variable the content of the IRAT coordination message.

Also assigned following :

UTRAN.receive (cmr_PDP_Established) -> value v_IRAT_Coordination_MSG;   
v_PDPEstEarlier := v_IRAT_Coordination_MSG.PdpEstEarlier;
Passed the variable v_PDPEstEarlier to the function f_EUTRA_IdleMode_TAU_UponMobility_FromOtherRAT.
Based upon variable v_PDPEstEarlier called function f_EUTRA_TrackingAreaUpdateFromAnotherRAT with different set of arguments. This is done so that the non-IMS path does not get affected.
Please see screenshot.  

	Module
	CSG_TestcasesUG.ttcn

	Label
	WA#


Before:
	function f_TC_6_3_3_EUTRA() runs on EUTRA_PTC
  {
    var template (value) CellPowerList_Type v_CellPowerList_T0toT1;
    var template (value) CellPowerList_Type v_CellPowerList_T1toT2;
    var template (value) CellPowerList_Type v_CellPowerList_T2toT3;
    var template (value) CellPowerList_Type v_CellPowerList_T3toT4;
    var template (value) CellPowerList_Type v_CellPowerList_Postamble;
    var IRAT_CoOrd_SysInfo_Type v_UTRANSysInfo;
    var GutiParameters_Type v_GutiParams;
    var CSG_Identity v_CSG_Identity_Cell2 := '000000000000000000000000001'B;
    var CellReselectionPriority v_CellReselectionPriority_Cell5 := 5;
    var boolean v_CSInOtherRAT := false;
    var IRAT_Coordination_MSG v_IRAT_Coordination_MSG;
    var template (value) SystemInformationBlockType6 v_Sib6;        
    timer t_WaitTime := tsc_IdleMode_GenericTimer;
    v_CellPowerList_T0toT1 := {
      cs_CellPower(eutra_Cell1, -91),
      cs_CellPower(eutra_Cell2, tsc_ServingCellRS_EPRE)
    };
    v_CellPowerList_T1toT2 := {
      cs_CellPower(eutra_Cell1, tsc_NonSuitableOffCellRS_EPRE),
      cs_CellPower(eutra_Cell2, tsc_NonSuitableOffCellRS_EPRE)
    };
    v_CellPowerList_T2toT3 := {
      cs_CellPower(eutra_Cell1, tsc_ServingCellRS_EPRE)
    };
    v_CellPowerList_T3toT4 := {
      cs_CellPower(eutra_Cell1, -91),
      cs_CellPower(eutra_Cell2, tsc_ServingCellRS_EPRE)
    };
    v_CellPowerList_Postamble := {
      cs_CellPower(eutra_Cell1, tsc_NonSuitableOffCellRS_EPRE)
    };
    //Initialise all cells, security and mobile parameters
    f_EUTRA_Init(c11);
    //Set system information combination 4 in cell 1
    v_Sib6 := f_EUTRA_CellInfo_GetSib6(eutra_Cell1); //@sic R5s130922 sic@
    f_EUTRA_ChangeSysinfoCombination (eutra_Cell1, c4);
    f_EUTRA_CellInfo_SetSIB6(eutra_Cell1, v_Sib6);     //@sic R5s130922 sic@
    //Coordinate with UTRAN component to receive UTRAN frequencies
    v_UTRANSysInfo := f_IRAT_WaitForCoOrd_SysInfo (UTRAN);
    //Send the cell info to the UTRAN component
    f_IRAT_Send_EUTRA_SysInfoToOtherPort(UTRAN,
                                         f_EUTRA_CellInfo_GetEARFCN_DL(eutra_Cell1),
                                         f_ConvertDL_BandwidthToInteger (f_EUTRA_CellInfo_GetDL_ChBandwidth (eutra_Cell1)));
    //Set TAC according to 36.523-1 Table 6.0.4-2: Tracking Area Code (TAC) for E-UTRAN cells
    //TAC for Cell 1 is set to 1 in f_EUTRA_Init
    f_EUTRA_CellInfo_SetTAC (eutra_Cell2, tsc_IdleMode_TAC_Cell2);
    //Set Sys. info for Cell 2 according to Table 6.3.3.3.3-2
    f_EUTRA_CellInfo_SetSysInfo_Csg_Ind(eutra_Cell2, true);
    f_EUTRA_CellInfo_SetSysInfo_Csg_Id(eutra_Cell2, v_CSG_Identity_Cell2);
    //Set SIB4 for cell 2 according to Table 6.3.3.3.3-3 and according to default message contents defined in 36.508
    f_EUTRA_CellInfo_SetSIB4 (eutra_Cell2, cs_SIB4_IntraFreqNeighCellList_and_Csg_PhysCellIdRange(cs_IntraFreqNeighCellInfoList_Cell2_ThreeEntriesDef,
                                                                                                  cs_PhysCellIdRange(2, n4)));
    //Set SIB6 for cell 1 and 2 according to Table 6.3.3.3.3-5
    if (f_IRAT_GetUTRAN_FDD_TDD_Mode () == UTRAN_FDD) {
      f_EUTRA_SetSIB6_InterFreqCarrierFreqList_F8(eutra_Cell1,
                                                  cs_CarrierFreqUTRA_FDD_CRPrio (v_UTRANSysInfo.Utran[0].FDD.Freq, v_CellReselectionPriority_Cell5),
                                                  omit);
      f_EUTRA_SetSIB6_InterFreqCarrierFreqList_F8(eutra_Cell2,
                                                  cs_CarrierFreqUTRA_FDD_CRPrio (v_UTRANSysInfo.Utran[0].FDD.Freq, v_CellReselectionPriority_Cell5),
                                                  omit);
    }
    else {
      f_EUTRA_SetSIB6_InterFreqCarrierFreqList_F8(eutra_Cell1,
                                                  omit,
                                                  cs_CarrierFreqUTRA_TDD_CRPrio (v_UTRANSysInfo.Utran[0].TDD.Freq, v_CellReselectionPriority_Cell5));
      f_EUTRA_SetSIB6_InterFreqCarrierFreqList_F8(eutra_Cell2,
                                                  omit,
                                                  cs_CarrierFreqUTRA_TDD_CRPrio (v_UTRANSysInfo.Utran[0].TDD.Freq, v_CellReselectionPriority_Cell5));
    }
    //Create and configure Cells
    f_EUTRA_CellConfig_Def(eutra_Cell1);
    f_EUTRA_CellConfig_Def(eutra_Cell2);
    f_UT_USIM_Insert (UT, "USIM with service n86 available");
    // The UEs allowed CSG list shall not contain CSG ID 1
    // Remove CSG ID 1 from UE's Allowed CSG list
    // @sic R5-153578 R5s150803 sic@
    f_EUTRA_EmptyACSGL(eutra_Cell2, {v_CSG_Identity_Cell2});
    // Broadcast the correct CSG id in eutra_Cell2
    f_EUTRA_CellInfo_SetSysInfo_Csg_Id(eutra_Cell2, v_CSG_Identity_Cell2);
    f_EUTRA_SS_ConfigureSysinfo(eutra_Cell2);  //@sic R5s130922 sic@
    //@sic R5s130922 sic@
    // Coordination to start preamble in UTRAN
    f_IRAT_SendCoOrd (UTRAN, cms_IRAT_Trigger);  //@sic R5s130922 sic@
    f_EUTRA_TestBody_Set (true);
    // Since Authentication parameters were updated in the idle updated procedure on UTRAN side...
    //...receive the updated parameters
    v_CSInOtherRAT := f_EUTRA_InterRAT_InitialiseAuthParams(UTRAN);
    UTRAN.receive (cmr_IRAT_OctetData) -> value v_IRAT_Coordination_MSG;
    //@siclog "Step 14" siclog@
    //Change Cell 1 and Cell 2 power levels according to the row "T1" in table 6.3.3.3.2-1.
    f_EUTRA_SetCellPowerList (v_CellPowerList_T0toT1);
    //@siclog "Step 15" siclog@
    //Perform manual CSG selection of cell 2
    v_GutiParams := f_EUTRA_CellInfo_GetGuti(eutra_Cell2);
    f_UT_ManualPLMN_CSG(UT, f_Asn2Nas_PlmnId(v_GutiParams.PLMN_Identity), v_CSG_Identity_Cell2);
    //@siclog "Step 16" siclog@
    //Check that UE camps on Cell 2 with the generic test procedure TS 36.508 table 6.4.2.7A-2
    f_EUTRA_IdleMode_TAU_UponMobility_FromOtherRAT(eutra_Cell2, 
                                                    v_CSInOtherRAT, 
                                                    false, 
                                                    false,
                                                    v_IRAT_Coordination_MSG.OctetData); 
    //@siclog "Step 16A" siclog@
    //Wait 6 seconds for UE to receive system information
    f_Delay (6.0);
    //@siclog "Step 17" siclog@
    //Change Cell 1, Cell 2, Cell 5 and Cell 7 levels according to row "T2" in table 6.3.3.3.2-1 and table 6.3.3.3.2-2.
    f_EUTRA_SendAuthParameters(eutra_Cell2, UTRAN);
    f_EUTRA_SetCellPowerList (v_CellPowerList_T1toT2);
    f_IRAT_WaitForCoOrd_Trigger (UTRAN);
    //@siclog "Step 28" siclog@
    //Change Cell 1, Cell 2, Cell 5 and Cell 7 levels according to row "T3" in table 6.3.3.3.2-1 and table 6.3.3.3.2-2.
    f_EUTRA_SetCellPowerList (v_CellPowerList_T2toT3);
    v_CSInOtherRAT := f_EUTRA_InterRAT_InitialiseAuthParams(UTRAN);
    //@siclog "Step 39" siclog@
    //Change Cell 1 and Cell 2 level according to row "T4" in table 6.3.3.3.2-1.
    f_EUTRA_SetCellPowerList (v_CellPowerList_T3toT4);
    t_WaitTime.start;
    //@siclog "Step 40-46" siclog@
    //Check that UE camps on Cell 1 with the generic test procedure TS 36.508 subclause 6.4.2.7A
    f_EUTRA_TrackingAreaUpdateFromAnotherRAT(eutra_Cell2, 
                                                tsc_RRC_TI_Def, 
                                                v_CSInOtherRAT, 
                                                false, 
                                                true, 
                                                v_IRAT_Coordination_MSG.OctetData, 
                                                noRrcConnectionRelease, 
                                                NORMAL); //@sic R5s130922, R5s150830 sic@    
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 6.3.3 Step 46");
    t_WaitTime.stop; // If this times out it will be picked up in the default
    f_EUTRA_TestBody_Set (false);
    //Switch all other cells off
    f_IRAT_SendCoOrd (UTRAN, cms_IRAT_Trigger);
    f_EUTRA_SetCellPowerList(v_CellPowerList_Postamble);
    f_EUTRA_UE_Detach_SwitchOffUe(eutra_Cell2, RRC_CONNECTED, NORMAL, omit);//@sic R5s150803 R5s160091 sic@
    f_EUTRA_Remove_CSG_From_ACSGL ( eutra_Cell2, v_CSG_Identity_Cell2);//@sic R5s150803 sic@
    f_EUTRA_ReleaseAllCells();//@sic R5s150803 sic@
  }


After:
	function f_TC_6_3_3_EUTRA() runs on EUTRA_PTC
  {
    var template (value) CellPowerList_Type v_CellPowerList_T0toT1;
    var template (value) CellPowerList_Type v_CellPowerList_T1toT2;
    var template (value) CellPowerList_Type v_CellPowerList_T2toT3;
    var template (value) CellPowerList_Type v_CellPowerList_T3toT4;
    var template (value) CellPowerList_Type v_CellPowerList_Postamble;
    var IRAT_CoOrd_SysInfo_Type v_UTRANSysInfo;
    var GutiParameters_Type v_GutiParams;
    var CSG_Identity v_CSG_Identity_Cell2 := '000000000000000000000000001'B;
    var CellReselectionPriority v_CellReselectionPriority_Cell5 := 5;
    var boolean v_CSInOtherRAT := false;
    var IRAT_Coordination_MSG v_IRAT_Coordination_MSG;
    var template (value) SystemInformationBlockType6 v_Sib6;
    var boolean v_PDPEstEarlier := false;//WA#6_3_3    
    timer t_WaitTime := tsc_IdleMode_GenericTimer;
    v_CellPowerList_T0toT1 := {
      cs_CellPower(eutra_Cell1, -91),
      cs_CellPower(eutra_Cell2, tsc_ServingCellRS_EPRE)
    };
    v_CellPowerList_T1toT2 := {
      cs_CellPower(eutra_Cell1, tsc_NonSuitableOffCellRS_EPRE),
      cs_CellPower(eutra_Cell2, tsc_NonSuitableOffCellRS_EPRE)
    };
    v_CellPowerList_T2toT3 := {
      cs_CellPower(eutra_Cell1, tsc_ServingCellRS_EPRE)
    };
    v_CellPowerList_T3toT4 := {
      cs_CellPower(eutra_Cell1, -91),
      cs_CellPower(eutra_Cell2, tsc_ServingCellRS_EPRE)
    };
    v_CellPowerList_Postamble := {
      cs_CellPower(eutra_Cell1, tsc_NonSuitableOffCellRS_EPRE)
    };
    //Initialise all cells, security and mobile parameters
    f_EUTRA_Init(c11);
    //Set system information combination 4 in cell 1
    v_Sib6 := f_EUTRA_CellInfo_GetSib6(eutra_Cell1); //@sic R5s130922 sic@
    f_EUTRA_ChangeSysinfoCombination (eutra_Cell1, c4);
    f_EUTRA_CellInfo_SetSIB6(eutra_Cell1, v_Sib6);     //@sic R5s130922 sic@
    //Coordinate with UTRAN component to receive UTRAN frequencies
    v_UTRANSysInfo := f_IRAT_WaitForCoOrd_SysInfo (UTRAN);
    //Send the cell info to the UTRAN component
    f_IRAT_Send_EUTRA_SysInfoToOtherPort(UTRAN,
                                         f_EUTRA_CellInfo_GetEARFCN_DL(eutra_Cell1),
                                         f_ConvertDL_BandwidthToInteger (f_EUTRA_CellInfo_GetDL_ChBandwidth (eutra_Cell1)));
    //Set TAC according to 36.523-1 Table 6.0.4-2: Tracking Area Code (TAC) for E-UTRAN cells
    //TAC for Cell 1 is set to 1 in f_EUTRA_Init
    f_EUTRA_CellInfo_SetTAC (eutra_Cell2, tsc_IdleMode_TAC_Cell2);
    //Set Sys. info for Cell 2 according to Table 6.3.3.3.3-2
    f_EUTRA_CellInfo_SetSysInfo_Csg_Ind(eutra_Cell2, true);
    f_EUTRA_CellInfo_SetSysInfo_Csg_Id(eutra_Cell2, v_CSG_Identity_Cell2);
    //Set SIB4 for cell 2 according to Table 6.3.3.3.3-3 and according to default message contents defined in 36.508
    f_EUTRA_CellInfo_SetSIB4 (eutra_Cell2, cs_SIB4_IntraFreqNeighCellList_and_Csg_PhysCellIdRange(cs_IntraFreqNeighCellInfoList_Cell2_ThreeEntriesDef,
                                                                                                  cs_PhysCellIdRange(2, n4)));
    //Set SIB6 for cell 1 and 2 according to Table 6.3.3.3.3-5
    if (f_IRAT_GetUTRAN_FDD_TDD_Mode () == UTRAN_FDD) {
      f_EUTRA_SetSIB6_InterFreqCarrierFreqList_F8(eutra_Cell1,
                                                  cs_CarrierFreqUTRA_FDD_CRPrio (v_UTRANSysInfo.Utran[0].FDD.Freq, v_CellReselectionPriority_Cell5),
                                                  omit);
      f_EUTRA_SetSIB6_InterFreqCarrierFreqList_F8(eutra_Cell2,
                                                  cs_CarrierFreqUTRA_FDD_CRPrio (v_UTRANSysInfo.Utran[0].FDD.Freq, v_CellReselectionPriority_Cell5),
                                                  omit);
    }
    else {
      f_EUTRA_SetSIB6_InterFreqCarrierFreqList_F8(eutra_Cell1,
                                                  omit,
                                                  cs_CarrierFreqUTRA_TDD_CRPrio (v_UTRANSysInfo.Utran[0].TDD.Freq, v_CellReselectionPriority_Cell5));
      f_EUTRA_SetSIB6_InterFreqCarrierFreqList_F8(eutra_Cell2,
                                                  omit,
                                                  cs_CarrierFreqUTRA_TDD_CRPrio (v_UTRANSysInfo.Utran[0].TDD.Freq, v_CellReselectionPriority_Cell5));
    }
    //Create and configure Cells
    f_EUTRA_CellConfig_Def(eutra_Cell1);
    f_EUTRA_CellConfig_Def(eutra_Cell2);
    f_UT_USIM_Insert (UT, "USIM with service n86 available");
    // The UEs allowed CSG list shall not contain CSG ID 1
    // Remove CSG ID 1 from UE's Allowed CSG list
    // @sic R5-153578 R5s150803 sic@
    f_EUTRA_EmptyACSGL(eutra_Cell2, {v_CSG_Identity_Cell2});
    // Broadcast the correct CSG id in eutra_Cell2
    f_EUTRA_CellInfo_SetSysInfo_Csg_Id(eutra_Cell2, v_CSG_Identity_Cell2);
    f_EUTRA_SS_ConfigureSysinfo(eutra_Cell2);  //@sic R5s130922 sic@
    //@sic R5s130922 sic@
    // Coordination to start preamble in UTRAN
    f_IRAT_SendCoOrd (UTRAN, cms_IRAT_Trigger);  //@sic R5s130922 sic@
    f_EUTRA_TestBody_Set (true);
    // Since Authentication parameters were updated in the idle updated procedure on UTRAN side...
    //...receive the updated parameters
    v_CSInOtherRAT := f_EUTRA_InterRAT_InitialiseAuthParams(UTRAN);
    UTRAN.receive (cmr_PDP_Established) -> value v_IRAT_Coordination_MSG; //WA#6_3_3     
    v_PDPEstEarlier := v_IRAT_Coordination_MSG.PdpEstEarlier;//WA#6_3_3
    UTRAN.receive (cmr_IRAT_OctetData) -> value v_IRAT_Coordination_MSG;
    //@siclog "Step 14" siclog@
    //Change Cell 1 and Cell 2 power levels according to the row "T1" in table 6.3.3.3.2-1.
    f_EUTRA_SetCellPowerList (v_CellPowerList_T0toT1);
    //@siclog "Step 15" siclog@
    //Perform manual CSG selection of cell 2
    v_GutiParams := f_EUTRA_CellInfo_GetGuti(eutra_Cell2);
    f_UT_ManualPLMN_CSG(UT, f_Asn2Nas_PlmnId(v_GutiParams.PLMN_Identity), v_CSG_Identity_Cell2);
    //@siclog "Step 16" siclog@
    //Check that UE camps on Cell 2 with the generic test procedure TS 36.508 table 6.4.2.7A-2
    f_EUTRA_IdleMode_TAU_UponMobility_FromOtherRAT(eutra_Cell2, 
                                                    v_CSInOtherRAT, 
                                                    false, 
                                                    false,
                                                    v_IRAT_Coordination_MSG.OctetData,
                                                    -,//WA#6_3_3
                                                    v_PDPEstEarlier/*WA#6_3_3*/); 
    //@siclog "Step 16A" siclog@
    //Wait 6 seconds for UE to receive system information
    f_Delay (6.0);
    //@siclog "Step 17" siclog@
    //Change Cell 1, Cell 2, Cell 5 and Cell 7 levels according to row "T2" in table 6.3.3.3.2-1 and table 6.3.3.3.2-2.
    f_EUTRA_SendAuthParameters(eutra_Cell2, UTRAN);
    f_EUTRA_SetCellPowerList (v_CellPowerList_T1toT2);
    f_IRAT_WaitForCoOrd_Trigger (UTRAN);
    //@siclog "Step 28" siclog@
    //Change Cell 1, Cell 2, Cell 5 and Cell 7 levels according to row "T3" in table 6.3.3.3.2-1 and table 6.3.3.3.2-2.
    f_EUTRA_SetCellPowerList (v_CellPowerList_T2toT3);
    v_CSInOtherRAT := f_EUTRA_InterRAT_InitialiseAuthParams(UTRAN);
    //@siclog "Step 39" siclog@
    //Change Cell 1 and Cell 2 level according to row "T4" in table 6.3.3.3.2-1.
    f_EUTRA_SetCellPowerList (v_CellPowerList_T3toT4);
    t_WaitTime.start;
    //@siclog "Step 40-46" siclog@
    //Check that UE camps on Cell 1 with the generic test procedure TS 36.508 subclause 6.4.2.7A
    if(v_PDPEstEarlier == true){
        f_EUTRA_TrackingAreaUpdateFromAnotherRAT(eutra_Cell2, 
                                                tsc_RRC_TI_Def, 
                                                v_CSInOtherRAT, 
                                                false, 
                                                false,//WA#6_3_3 true, 
                                                v_IRAT_Coordination_MSG.OctetData, 
                                                noRrcConnectionRelease, 
                                                NORMAL,
                                                v_PDPEstEarlier/*WA#6_3_3*/); //@sic R5s130922, R5s150830 sic@
    }
    else{
        f_EUTRA_TrackingAreaUpdateFromAnotherRAT(eutra_Cell2, 
                                                tsc_RRC_TI_Def, 
                                                v_CSInOtherRAT, 
                                                false, 
                                                true, 
                                                v_IRAT_Coordination_MSG.OctetData, 
                                                noRrcConnectionRelease, 
                                                NORMAL,
                                                v_PDPEstEarlier/*WA#6_3_3*/); //@sic R5s130922, R5s150830 sic@
    }
    f_EUTRA_PreliminaryPass(__FILE__, __LINE__, "Test Case 6.3.3 Step 46");
    t_WaitTime.stop; // If this times out it will be picked up in the default
    f_EUTRA_TestBody_Set (false);
    //Switch all other cells off
    f_IRAT_SendCoOrd (UTRAN, cms_IRAT_Trigger);
    f_EUTRA_SetCellPowerList(v_CellPowerList_Postamble);
    f_EUTRA_UE_Detach_SwitchOffUe(eutra_Cell2, RRC_CONNECTED, NORMAL, omit);//@sic R5s150803 R5s160091 sic@
    f_EUTRA_Remove_CSG_From_ACSGL ( eutra_Cell2, v_CSG_Identity_Cell2);//@sic R5s150803 sic@
    f_EUTRA_ReleaseAllCells();//@sic R5s150803 sic@
  }


